The Second Set of Magic Angles of Projectile Motion

Haiduke Sarafian

By suppressing the kinematics of projectile motion, we show that a projectile thrown at 49.50° above the horizontal maximizes the trajectory's perimeter. We also identify two unique projectile angles, 15.5° and 88.9°, that give two different parabolas with identical perimeters and identical areas.

■ The Magic Angles of Projectile Motion

A projectile thrown from the ground in a vacuum with an initial speed v_0 at an angle θ above the horizontal has a trajectory of length $L(v_0, \theta)$ and encompasses an area Area(v_0, θ) [1].

$$L(v_0, \theta) = \frac{v_0^2}{g} \left\{\sin(\theta) - \cos^2(\theta) \ln\left[\frac{\cos(\theta)}{1+\sin(\theta)}\right]\right\},$$

$$\text{Area}(v_0, \theta) = \frac{2}{3} \left(\frac{v_0^2}{g}\right)^2 \cos(\theta) \sin^3(\theta).$$

The length can be rewritten using $\text{gd}^{-1}(\theta) = -\ln\left[\frac{\cos(\theta)}{1+\sin(\theta)}\right]$, an inverse Gudermannian function. In addition, a projectile has range

$$R(v_0, \theta) = \frac{v_0^2}{g} \sin(2\,\theta).$$

These equations share a peculiar property: the kinematics of the problem, the length $\frac{v_0^2}{g}$, is a factor separate from the parts that depend on the angle. This separation comes from the parametric representation of the projectile in Cartesian coordinates,

$$\{x(t),\ y(t)\} = \left\{v_0 \cos(\theta)\, t,\ -\frac{1}{2}\, g\, t^2 + v_0 \sin(\theta)\, t\right\}. \tag{1}$$

Setting the kinematic factor $K = \frac{v_0^2}{g}$ to 1 allows us to study the global features of the purely geometric aspects of projectile motion.

The given length, area, and range for $K = 1$ are

In[1]:=
```
length[θ_] := Sin[θ] - Cos[θ]^2 Log[Cos[θ]/(1 + Sin[θ])]
area[θ_] := 2/3 Cos[θ] Sin[θ]^3
range[θ_] := Sin[2 θ]
```

Here is the trajectory's perimeter.

In[4]:=
```
perimeter[θ_] := length[θ] + range[θ]
```

In[5]:=
```
Plot[{perimeter[π/180 t], area[π/180 t]}, {t, 0, 90},
 PlotStyle → {{}, {Dashing[{0.01}]}}, Frame → True,
 FrameLabel → {"θ", "perimeter, area", None, None},
 PlotRange → {{0, 90}, {0, 2.2}}, GridLines → Automatic]
```

From In[5]:=



Figure 1. The perimeter (solid line) and the area (dashed line) versus the initial angle θ.

Contrary to our intuition, the longest perimeter does not encompass the largest area; 60° is the angle that maximizes the area.

In[6]:=
```
Select[180/π θ /. Solve[D[area[θ], θ] == 0, θ], 0 < # < 90 &]
```

```
Solve::ifun :
 Inverse functions are being used by Solve, so some solutions may not
   be found; use Reduce for complete solution information. More …
```

Out[6]= {60}

Similarly, to determine the angle (in degrees) that maximizes the perimeter, we set the slope of the perimeter to zero and solve the transcendental equation.

In[7]:=
```
180/π θ /. FindRoot[D[perimeter[θ], θ] == 0, {θ, 0, 1}]
```

Out[7]= 49.5097

In[8]:=
```
ParametricPlot[{area[θ], perimeter[θ]}, {θ, 0, π/2},
  Frame → True, FrameLabel → {"area", "perimeter", None, None},
  PlotRange → {{0, 0.23}, {0, 2.3}}, GridLines → Automatic]
```

From In[8]:=



Figure 2. The perimeter versus the area for projectiles of initial angles $0 \le \theta \le 90°$.

The point where the curve crosses itself determines two angles at which the corresponding pairs of areas and perimeters are the same. To evaluate these *magic* angles, we solve the simultaneous equations.

In[9]:=
```
{α, β} = 180/π {θ1, θ2} /.
   FindRoot[{area[θ1] == area[θ2], perimeter[θ1] == perimeter[θ2]},
    {{θ1, 0}, {θ2, .9 Pi / 2}},
    WorkingPrecision → 1.01 MachinePrecision] // Chop
```

Out[9]=
```
{15.52988588878259, 88.9398614918464}
```

To plot the unique pair of corresponding kinematic-independent parabolic trajectories, we eliminate t between $\{x(t),\ y(t)\}$ as given in equation (1).

In[10]:=
```
y[θ_, x_] :=
  y /. Solve[Eliminate[{x == v0 Cos[θ] t, y == -g t^2/2 + v0 Sin[θ] t,
      v0^2/g == 1}, {t, v0, g}], y][[1]]
```

In[11]:=
```
y[θ, x]
```

Out[11]=

$$-\frac{1}{2}\,\mathrm{Sec}[\theta]^2\,(x^2 - 2\,x\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta])$$

In[12]:=
```
Needs["Graphics`FilledPlot`"];
```

In[13]:=
```
FilledPlot[
 {y[α Degree, x], y[β Degree, x]}, {x, 0, range[α Degree]},
 Fills → {{{1, Axis}, GrayLevel[0.7]}, {{2, Axis}, GrayLevel[0.5]}},
 Curves → Front, Frame → True, FrameLabel → {"x", "y", None, None},
 PlotRange → {{0, 0.52}, {0, 0.52}}, GridLines → Automatic]
```

From In[13]:=



Figure 3. The magic pair of kinematic-independent trajectories. These two parabolas have the same perimeters and the same areas.

■ Acknowledgment

The author would like to thank the referee's useful suggestions.

■ References

[1] H. Sarafian, "The Magic Angles of Projectile Motion," *Mathematica in Education and Research*, **9** (3-4), 2000 pp. 20–26.

About the Author

Haiduke Sarafian is professor of physics at the York campus of the Commonwealth College of The Pennsylvania State University. He received his Ph.D. in theoretical nuclear physics from Michigan State University in 1983. He has also been a research associate at the National Superconducting Cyclotron Laboratory in Michigan and a visiting faculty member at Tokyo Metropolitan University in Japan and the University of Valencia in Spain. In 1999, he received a *Mathematica* Visiting Scholar Grant and is an independent *Mathematica* trainer (www.wolfram.com/services/training/sarafian.html).

Haiduke Sarafian
Commonwealth College
The Pennsylvania State University
Department of Physics
1031 Edgecomb Avenue
York, PA 17403
has2@psu.edu